
***************************************************************************************************************
/*
THIS DO-FILE:
Gets 2012 (wave 11) helper information from raw HRS data
*/
***************************************************************************************************************

clear
set more off
set maxvar 10000

***************************************************************************************************************
use "$rawdata\h12core\h12sta\H12G_R.DTA"
merge 1:m HHID PN using "$rawdata\h12core\h12sta\H12G_HP.DTA", keep(3)

egen LOPN=concat(NSUBHH OPN)

destring HHID, replace
destring PN, replace
gen double hhidpn=1000*HHID+PN

destring OPN, replace
destring LOPN, replace

gen double kidid=10000*HHID+LOPN

*helper type
gen helper_spouse=inlist(NG069,2,26,27)
gen helper_child=inlist(NG069,3,4,5,6,7,8,28,30,31,90,91)
gen helper_othfam=inlist(NG069,9,10,11,12,13,14,15,16,17,18,19,33)
gen helper_nonfam=inlist(NG069,20,21,22,23,24,25)
foreach var in helper_spouse helper_nonfam helper_child helper_othfam {
	replace `var'=. if helper_spouse==0 & helper_nonfam==0 & helper_child==0 & helper_othfam==0
}
*1 out of 6735 are missing -- they won't be counted as helpers

*helper paid?
gen 	helper_paid=(NG076==1) if inlist(NG076,1,5) & helper_nonfam==1
replace helper_paid=1 if NG069==22
						 
*helper frequency - converted to days per month
gen 	helper_dpm=NG070 if NG070<=31  //reported days per month
replace helper_dpm=NG071*4 if NG071<=7 //reported days per week
replace helper_dpm=30 if NG072==1	   //reported every day
replace helper_dpm=0 if NG070==96 	   //not a helper last month
*impute the "DK" responses with median for that type of helper
forval x=2/91 {
	qui sum helper_dpm if NG069==`x', det
	qui replace helper_dpm=r(p50) if NG069==`x' & (inlist(NG070,98,99) | NG071==8 | inlist(NG072,8,9)) & helper_dpm==.
}
qui sum helper_dpm, det

*helper hours per day
gen 	helper_hpd=NG073 if NG073<=24
*impute the "DK" responses by family/other and days per month categories
foreach var in helper_spouse helper_child helper_othfam helper_nonfam {
	sum helper_hpd if `var'==1 & helper_dpm>=1 & helper_dpm<=7, det
	replace helper_hpd=r(p50) if `var'==1 & helper_dpm>=1 & helper_dpm<=7 & inlist(NG073,98,99) & helper_hpd==.
	sum helper_hpd if `var'==1 & helper_dpm>=8 & helper_dpm<=27, det
	replace helper_hpd=r(p50) if `var'==1 & helper_dpm>=8 & helper_dpm<=27 & inlist(NG073,98,99) & helper_hpd==.	
	sum helper_hpd if `var'==1 & helper_dpm>=28 & helper_dpm<=31, det
	replace helper_hpd=r(p50) if `var'==1 & helper_dpm>=28 & helper_dpm<=31 & inlist(NG073,98,99) & helper_hpd==.
}

*hours per month
gen		helper_hpm=helper_hpd*helper_dpm
*those who report 0 or . dpm or hpd won't be counted as helpers unless AL/NH helper

*type of care
egen helper_fam=rowtotal(helper_spouse helper_child helper_othfam), missing
bys hhidpn: egen rec_inf=max(helper_fam)
bys hhidpn: egen rec_form=max(helper_nonfam)
bys hhidpn: egen rec_paid=max(helper_paid)

*hours - total over all of person's helpers
gen hrs_inf_temp=helper_hpm if helper_fam==1
bys hhidpn: egen hrs_inf=sum(hrs_inf_temp)
gen hrs_form_temp=helper_hpm if helper_nonfam==1
bys hhidpn: egen hrs_form=sum(hrs_form_temp)
egen hrs_total=rowtotal(hrs_inf hrs_form)

***************************************************************************************************************
*ADL AND IADL (including money) HELPERS (up to 7)
preserve
drop if NADLNDX==. & NIADLNDX==. & NMNYNDX==. 
keep hhidpn OPN LOPN kidid helper_hpm helper_spouse helper_child helper_othfam helper_nonfam helper_paid ///
	 rec_inf rec_form hrs_inf hrs_form hrs_total
foreach var in helper_hpm helper_spouse helper_child helper_othfam helper_nonfam helper_paid rec_inf rec_form hrs_inf hrs_form hrs_total {
	rename `var' `var'11
}
duplicates drop
save "allADLIADLhelpers_11.dta", replace
restore
***************************************************************************************************************
